- **Zamiana zmiennych w operacjach ró∎niczkowych**

```
{x, x, x} /. x :> RandomReal[]
```

```
{0.737195, 0.374438, 0.665781}
```

Jesli wprowadzimy funkcje zlozona f(g(x))), to:

```
abc = f''[x] /. f → (f[g[#]] &)
```

$g'[x]^2 f''[g[x]] + f'[g[x]] g''[x]$

Tu podstawiajac za g konkretna wartosc, liczy dobrze:

```
abc /. {g → (2 * #^2 &)}
```

$4 f'[2 x^2] + 16 x^2 f''[2 x^2]$

Podobnie

```
Clear[f, g, x, abc]
abc[x_] := f''[x] /. f -> (f[g[#]] &)
abc[x] /. {g → (2 * #^2 &)}
```

$4 f'[2 x^2] + 16 x^2 f''[2 x^2]$

- **cwiczenie: zrobic to samo ale dla dwoch zmiennych**

```
D[f[#1, #2], #1]
```

$f^{(1,0)}[\#1, \#2]$

```
(D[f[#1, #2], #1]) /. f → (f[g[#1] * #2] &)
```

$\#2\, f'[g[\#1]\, \#2]\, g'[\#1]$

```
(* dobrze, bo k(x,y)=y*g[x], df[x,y]/dx=df/dk dk/dx=df/dk* y*g'[x] *)
```

Zadanie: dla funkcji f[x,y] zamieniamy zmienne (x,y) -> (α,β). Oblicz ∂f/∂x=∂f/$\partial\alpha$ $\partial\alpha$/∂x + ∂f/$\partial\beta$ $\partial\beta$/∂x . Oraz ∂^2 f/∂x^2. Policz ∂f/∂x dla α= x+y, β=x-y oraz f[α, β] = 4 α^2 + 5 β.

```
Clear[x, y, α, β]
last = D[f[#1, #2], #1] /. f → (f[α[#1, #2], β[#1, #2]] &)
```

$f^{(1,0)}[\alpha[\#1, \#2], \beta[\#1, \#2]]\, \alpha^{(1,0)}[\#1, \#2] + f^{(0,1)}[\alpha[\#1, \#2], \beta[\#1, \#2]]\, \beta^{(1,0)}[\#1, \#2]$

```
(* niech α=x+y, β=x-y, to: *)
last /. {α → (#1 + #2 &), β → (#1 - #2 &)}
```

$f^{(0,1)}[\#1 + \#2, \#1 - \#2] + f^{(1,0)}[\#1 + \#2, \#1 - \#2]$

```
(* niech f[α,β]=4α^2+5β, to *)
% /. {f → (2 * #1^2 + 5 * #2 &)}
```

$5 + 4 (\#1 + \#2)$

```
expr = a * F1'[r] * F2[r] + b * F1''[r]
expr /. {F1 → (F1[1 / #] &), F2 → (F2[1 / #] &)}
```

$a\, F2[r]\, F1'[r] + b\, F1''[r]$

$$-\frac{a\, F2[\frac{1}{r}]\, F1'[\frac{1}{r}]}{r^2} + b \left(\frac{2\, F1'[\frac{1}{r}]}{r^3} + \frac{F1''[\frac{1}{r}]}{r^4}\right)$$

```
expr = a * F1'[r] * F2[r] + b * F1''[r]
expr /. {F1 → (F1[1 / #] &), F2 → (F2[1 / #] &)} /. r -> 1 / x
```

$$a\,F2[r]\,F1'[r] + b\,F1''[r]$$

$$-a\,x^2\,F2[x]\,F1'[x] + b\left(2\,x^3\,F1'[x] + x^4\,F1''[x]\right)$$

- **Rozwiazmy konkretne rownanie przez zamiane zmiennych:**

Rozwi▮▮ równanie x*u_x +y*u_y = u stosujac zamiane ξ=x, η=y/x:

```
eqn = x * D[u[x, y], x] + y * D[u[x, y], y] == u[x, y]
test = D[u[x, y], x]
```

$$y\,u^{(0,1)}[x, y] + x\,u^{(1,0)}[x, y] == u[x, y]$$

$$u^{(1,0)}[x, y]$$

```
last1 = test /. {u → (u[ξ[#1, #2], η[#1, #2]] &)}
last2 = eqn /. {u → (u[ξ[#1, #2], η[#1, #2]] &)}
```

$$u^{(0,1)}[\xi[x, y], \eta[x, y]]\,\eta^{(1,0)}[x, y] + u^{(1,0)}[\xi[x, y], \eta[x, y]]\,\xi^{(1,0)}[x, y]$$

$$y\left(u^{(0,1)}[\xi[x, y], \eta[x, y]]\,\eta^{(0,1)}[x, y] + \xi^{(0,1)}[x, y]\,u^{(1,0)}[\xi[x, y], \eta[x, y]]\right) + x\left(u^{(0,1)}[\xi[x, y], \eta[x, y]]\,\eta^{(1,0)}[x, y] + u^{(1,0)}[\xi[x, y], \eta[x, y]]\,\xi^{(1,0)}[x, y]\right) == u[\xi[x, y], \eta[x, y]]$$

Widac, ze wymagane za transformacje zmiennych nowa=F(stare)

```
(*NIE:*) last /. {ξ :> x, η :> y / x}
```

$$u^{(1,0)}\left[x[x, y], \frac{y}{x}[x, y]\right] x^{(1,0)}[x, y] + u^{(0,1)}\left[x[x, y], \frac{y}{x}[x, y]\right] \left(\frac{y}{x}\right)^{(1,0)}[x, y]$$

a raczej

```
last1 /. {ξ → (#1 &), η → (#2 / #1 &)}
last2 = last2 /. {ξ → (#1 &), η → (#2 / #1 &)}
```

$$-\frac{y\,u^{(0,1)}\left[x, \frac{y}{x}\right]}{x^2} + u^{(1,0)}\left[x, \frac{y}{x}\right]$$

$$\frac{y\,u^{(0,1)}\left[x, \frac{y}{x}\right]}{x} + x\left(-\frac{y\,u^{(0,1)}\left[x, \frac{y}{x}\right]}{x^2} + u^{(1,0)}\left[x, \frac{y}{x}\right]\right) == u\left[x, \frac{y}{x}\right]$$

```
last2 = Simplify[last2]
```

$$u\left[x, \frac{y}{x}\right] == x\,u^{(1,0)}\left[x, \frac{y}{x}\right]$$

```
Solve[{ξ == x, η == y / x}, {x, y}]
```

$$\{\{y \to \eta\,\xi, x \to \xi\}\}$$

```
last2 /. {x → ξ, y → ξ * η}
```

$$u[\xi, \eta] == \xi\,u^{(1,0)}[\xi, \eta]$$